

## SOLUCIÓN EJERCICIOS DE FUNCIONES ESCALARES

### BD COMPRAVENTA

1.- Función que acepte el identificador de un pedido y devuelva el valor del pedido **sin**/con cargo.

```
ALTER FUNCTION ValorPedidoSin
(@idpedido int)
RETURNS DECIMAL(10,4)
AS
BEGIN
    RETURN (SELECT SUM(PrecioUnidad*Cantidad*(1-Descuento))
            FROM [Detalles de Pedidos]
            WHERE IdPedido=@IdPedido
            GROUP BY IdPedido)
END
```

```
CREATE FUNCTION ValorPedidoCon
(@idpedido int)
RETURNS DECIMAL(10,4)
AS
BEGIN
    RETURN dbo.ValorPedidoSin(@IdPedido)+
           (SELECT Cargo
            FROM Pedidos
            WHERE IdPedido=@IdPedido)
END
```

```
SELECT IdPedido,dbo.ValorPedidoSin(IdPedido)ValorSin,
       dbo.ValorPedidoCon(Idpedido) ValorCon
FROM pedidos
```

2.- Función que acepta un año y devuelve el nombre del mejor cliente.  
El mejor cliente es el que ha realizado pedidos por un mayor valor en dicho año.

```
ALTER VIEW VMejorClienteAño
AS
    SELECT IdCliente, YEAR(FechaPedido) Año,
           SUM(dbo.ValorPedidoSin(IdPedido)) Valor
    FROM Pedidos
    GROUP BY IdCliente, YEAR(FechaPedido)
```

```

ALTER FUNCTION FMejorClienteAño
    (@Año int)
    RETURNS nvarchar(40)
AS
    BEGIN
        RETURN (SELECT NombreCompañía
                FROM Clientes
                WHERE IdCliente=(
                    SELECT TOP 1 IdCliente
                    FROM VMejorClienteAño
                    WHERE Año=@Año
                    ORDER BY Valor DESC ))
    END

```

```

SELECT dbo.FMejorClienteAño(1997)

```

3.- Función que acepta un mes y un año y visualiza el nombre del mejor vendedor en dicho mes de dicho año.

```

ALTER VIEW VMejorVendedorMes
AS
    SELECT IdEmpleado, YEAR(FechaPedido) Año,
           MONTH(FechaPedido) Mes,
           SUM(dbo.ValorPedidoSin(IdPedido)) Valor
    FROM Pedidos
    GROUP BY IdEmpleado, YEAR(FechaPedido),
            MONTH(FechaPedido)

```

```

ALTER FUNCTION FMejorVendedorMes
    (@Año int,
     @Mes int)
    RETURNS nvarchar(30)
AS
    BEGIN
        RETURN (SELECT Nombre+' '+Apellidos
                FROM Empleados
                WHERE IdEmpleado=(
                    SELECT TOP 1 IdEmpleado
                    FROM VMejorVendedorMes
                    WHERE Año=@Año AND Mes=@Mes
                    ORDER BY Valor DESC))
    END

```

```

SELECT dbo.FMejorVendedorMes(1998,5)

```

4.- Función que acepta el nombre de un producto y devuelve la fecha en que fue pedido por última vez.

```
ALTER VIEW VFechaPedidoProducto
AS
SELECT TOP 2155 IdProducto, FechaPedido
FROM Pedidos P JOIN
    [Detalles de Pedidos] DDP
ON P.IdPedido=DDP.IdPedido
```

```
ALTER FUNCTION FechaPedidoProducto
(@Producto nvarchar(40))
RETURNS datetime
AS
BEGIN
    RETURN (
        SELECT TOP 1 FechaPedido
        FROM VFechaPedidoProducto
        WHERE IdProducto=(
            SELECT IdProducto
            FROM Productos
            WHERE NombreProducto =@Producto)
        ORDER BY FechaPedido Desc)
END
```

```
SELECT dbo.Formato_Fecha(dbo.FechaPedidoProducto('Especias Cajun del
chef Anton'),'')
```

5.- Función que acepta el nombre de un vendedor y devuelve la fecha del día en que consiguió pedidos por un mayor valor.

```
CREATE VIEW VFechaPedidoVendedor
AS
SELECT IdEmpleado, FechaPedido, SUM(dbo.ValorPedidoSin(IdPedido))
ValorPedidos
FROM Pedidos
GROUP BY IdEmpleado, FechaPedido
```

```
ALTER FUNCTION FechaPedidoVendedor
(@Nombre nvarchar(10))
RETURNS datetime
AS
BEGIN
    RETURN (
        SELECT TOP 1 FechaPedido
        FROM VFechaPedidoVendedor
        WHERE IdEmpleado=(SELECT IdEmpleado
```

```

FROM Empleados
WHERE Nombre=@Nombre)
ORDER BY ValorPedidos DESC)
END

```

```

SELECT dbo.Formato_Fecha(dbo.FechaPedidoVendedor('Nancy'),'')

```

6.- Función que acepta dos fechas y el apellido de un empleado y devuelve el valor total de los pedidos (con descuentos y **sin** cargos) conseguidos por el empleado desde la fecha menor hasta la mayor, ambas inclusive. Las fechas deben poder teclearse en cualquier orden. Si **no** ha conseguido ningún pedido en dichas fechas deberá emitir el mensaje correspondiente.

```

ALTER FUNCTION PedidosEntreFechas
(@FechaInicio datetime,
 @FechaFin datetime,
 @Nombre nvarchar(10))
RETURNS varchar(50)
AS
BEGIN
    DECLARE @FechaCambio datetime
    IF @FechaInicio > @FechaFin
        BEGIN
            SET @FechaCambio=@FechaInicio
            SET @FechaInicio=@FechaFin
            SET @FechaFin=@FechaCambio
        END
    DECLARE @ValorPedidos decimal(10,4)
    SET @ValorPedidos = (
        SELECT SUM(dbo.ValorPedidoSin(IdPedido))
        FROM Pedidos
        WHERE FechaPedido BETWEEN @FechaInicio AND @FechaFin
        AND IdEmpleado=(SELECT IdEmpleado
                        FROM Empleados
                        WHERE Nombre=@nombre))
    DECLARE @mensaje varchar(50)
    IF @ValorPedidos IS NULL
        SET @mensaje='Ningún pedido'
    ELSE
        SET @mensaje=CONVERT(VARCHAR(40),@ValorPedidos)
    RETURN @mensaje
END

```

```

SELECT dbo.PedidosEntreFechas('01/01/97','31/01/97','Nancy')
SELECT dbo.PedidosEntreFechas('31/01/97','01/01/97','Nancy')
SELECT dbo.PedidosEntreFechas('18/07/96','19/07/96','Nancy')

```